Lezione 3 (parte prima)

Enrico Bertolazzi

```
> with(plots):
> # zero0(N,k) genera il k-esimo di N+1 nodi
  # equispaziati nell'intervallo [-1,1].
  zero0 := (N,k) -> 2*k/N - 1 ;
```

$$zero0 := (N,k) \rightarrow \frac{2k}{N} - 1 \quad (1)$$

```
> # esempio di uso
  seq(zero0(10,i),i=0..10) ;
```

$$-1, -\frac{4}{5}, -\frac{3}{5}, -\frac{2}{5}, -\frac{1}{5}, 0, \frac{1}{5}, \frac{2}{5}, \frac{3}{5}, \frac{4}{5}, 1 \quad (2)$$

```
> # zero1(N,k) genera il k-esimo di N+1 nodi
  # NON equispaziati nell'intervallo [-1,1].
  zero1 := (N,k) -> - cos(k*Pi/N) ;
```

$$zero1 := (N,k) \rightarrow -\cos\left(\frac{k\pi}{N}\right) \quad (3)$$

```
> # esempio di uso
  seq(zero1(4,i),i=0..4) ;
```

$$-1, -\frac{1}{2}\sqrt{2}, 0, \frac{1}{2}\sqrt{2}, 1 \quad (4)$$

```
> # zero2(N,k) genera il k-esimo di N+1 nodi
  # NON equispaziati nell'intervallo [-1,1].
  zero2 := (N,k) -> - cos(k*Pi/(N+1)+Pi/(2*(N+1))) ;
```

$$zero2 := (N,k) \rightarrow -\cos\left(\frac{k\pi}{N+1} + \frac{\pi}{2N+2}\right) \quad (5)$$

```
> # esempio di uso
  seq(zero2(3,i),i=0..3) ;
```

$$-\cos\left(\frac{1}{8}\pi\right), -\cos\left(\frac{3}{8}\pi\right), \cos\left(\frac{3}{8}\pi\right), \cos\left(\frac{1}{8}\pi\right) \quad (6)$$

> # $\omega(x) = \prod_{i=0}^{N}(x - x_i)$ dove $x_i = \frac{2i}{N} - 1$

```
OM0 := (N,x) -> mul(x-zero0(N,i),i=0..N) ;
```

$$OM0 := (N,x) \rightarrow mul(x - zero0(N,i), i = 0..N) \quad (7)$$

```
> # esempio di uso
  expand(OM0(5,x)) ;
```

(8)

$$x^6 - \frac{7}{5}x^4 + \frac{259}{625}x^2 - \frac{9}{625} \quad (8)$$

> # $\omega(x) = \prod_{i=0}^{N}(x - x_i)$ dove $x_i = -\cos\left(\frac{i\pi}{N}\right)$

```
OM1 := (N,x) -> mul(x-zero1(N,i),i=0..N) ;
```

$$OM1 := (N, x) \rightarrow mul(x - zero1(N, i), i = 0..N) \quad (9)$$

```
> # esempio di uso
  expand(OM1(4,x)) ;
```

$$x^5 - \frac{3}{2}x^3 + \frac{1}{2}x \quad (10)$$

> # $\omega(x) = \prod_{i=0}^{N}(x - x_i)$ dove $x_i = -\cos\left(\frac{i\pi}{N+1} + \frac{\pi}{2N+2}\right)$

```
OM2 := (N,x) -> mul(x-zero2(N,i),i=0..N) ;
```

$$OM2 := (N, x) \rightarrow mul(x - zero2(N, i), i = 0..N) \quad (11)$$

```
> # esempio di uso
  expand(OM2(3,x));
```

$$x^4 - x^2\cos\left(\frac{3}{8}\pi\right)^2 - \cos\left(\frac{1}{8}\pi\right)^2 x^2 + \cos\left(\frac{1}{8}\pi\right)^2\cos\left(\frac{3}{8}\pi\right)^2 \quad (12)$$

```
> # stampa dei polinomi per N=10
  N := 10 ;
  plot([OM0(N,x),OM1(N,x),OM2(N,x)],
       x=-1..1,y=-0.004..0.004,
       color=[black,red,blue]);
```

$$N := 10$$




```
> # TR trasforma da [-1,1] all'intervallo [a,b]
  TR := (a,b,x) -> (a+b)/2 + x *(b-a)/2 ;
```

$$TR := (a, b, x) \rightarrow \frac{1}{2}\, a + \frac{1}{2}\, b + \frac{1}{2}\, x\, (b - a) \qquad \textbf{(13)}$$

```
> # alcuni esempi
  TR(a,b,0) ; TR(a,b,-1) ; TR(a,b,1) ;
```

$$0 \qquad \textbf{(14)}$$

$$-5$$

$$5$$

```
> # calcolo i polinomi ''omega'' per l'intervallo [a,b]
  OMTR0 := (N,a,b,x) -> mul(x-TR(a,b,zero0(N,i)),i=0..N) ;
  OMTR1 := (N,a,b,x) -> mul(x-TR(a,b,zero1(N,i)),i=0..N) ;
  OMTR2 := (N,a,b,x) -> mul(x-TR(a,b,zero2(N,i)),i=0..N) ;
```

$$OMTR0 := (N, a, b, x) \rightarrow mul(x - TR(a, b, zero0(N, i)), i = 0 .. N) \qquad \textbf{(15)}$$

$$OMTR1 := (N, a, b, x) \rightarrow mul(x - TR(a, b, zero1(N, i)), i = 0 .. N)$$

$$OMTR2 := (N, a, b, x) \rightarrow mul(x - TR(a, b, zero2(N, i)), i = 0 .. N)$$

```
> # stampa dei polinomi per N=10
  N   := 15 ;
  a,b := -5,5 ;
  plot([OMTR0(N,a,b,x),OMTR1(N,a,b,x),OMTR2(N,a,b,x)],
       x=a..b,y=-5e7..5e7,
       color=[black,red,blue]);
```

$$N := 15$$

$$a, b := -5, 5$$




```
> # funzione per l'esempio di Runge
  f := x -> 1/(1+x^2) ;
```

$$f := x \to \frac{1}{1+x^2} \quad \textbf{(16)}$$

```
> # funzione e alcune sue derivate
  a,b := -5,5 ;
  plot([f(x),D(f)(x),(D@@2)(f)(x),(D@@8)(f)(x)],
       x=a..b,y=-2..2,
       color=[black,red,blue,green]);
```

$$a, b := -5, 5$$




```
> # polinomio interpolante per N=10 punti equispaziati
  N   := 10 :
  a,b := -5, 5:
  X   := [seq(TR(a,b,zero0(N,i)),i=0..N)]:
  Y   := [seq(f(X[i+1]),i=0..N)]:
  X   := evalf(X,40):
  Y   := evalf(Y,40):

  p0  := interp(X,Y,x) ;
  A   := plot([p0,f(x)],x=a..b,y=-1..2):
  B   := plot([seq([X[i],Y[i]],i=1..N+1)],style=POINT,symbolsize=20):
  display([A,B]);
```

$$p0 := -0.00002262443444\,x^{10} + 1.\,10^{-13}\,x^{9} + 1.000000000 + 0.001266968330\,x^{8} - 1.\,10^{-10}\,x + 3.\,10^{-12}\,x^{7} - 0.6742081460\,x^{2} - 0.02441176476\,x^{6} + 1.\,10^{-10}\,x^{3} + 6.\,10^{-11}\,x^{5} + 0.1973755662\,x^{4}$$




```
> # polinomio interpolante per N=10 punti NON equispaziati
  N   := 10 :
  a,b := -5, 5:
  X   := [seq(TR(a,b,zero1(N,i)),i=0..N)]:
  Y   := [seq(f(X[i+1]),i=0..N)]:
  X   := evalf(X,40):
  Y   := evalf(Y,40):
  p1  := interp(X,Y,x) ;

  A := plot([p1,f(x)],x=a..b,y=-1..2):
  B := plot([seq([X[i],Y[i]],i=1..N+1)],style=POINT,symbolsize=20):
  display([A,B]);
```

$$p1 := -0.000002873798799\,x^{10} - 1.36\;10^{-12}\,x^{9} + 1.000000000 + 0.0002184087020\,x^{8} - 1.21\;10^{-8}\,x + 7.10\;10^{-11}\,x^{7} - 0.4518664372\,x^{2} - 0.006168069931\,x^{6} + 7.34\;10^{-9}\,x^{3} - 1.26\;10^{-9}\,x^{5} + 0.07913561142\,x^{4}$$




```
> # polinomio interpolante per N=10 punti NON equispaziati
  N   := 10 :
  a,b := -5, 5:
  X   := [seq(TR(a,b,zero2(N,i)),i=0..N)]:
  Y   := [seq(f(X[i+1]),i=0..N)]:
  X   := evalf(X,40):
  Y   := evalf(Y,40):
  p2  := interp(X,Y,x) ;

  A := plot([p2,f(x)],x=a..b,y=-1..2):
  B := plot([seq([X[i],Y[i]],i=1..N+1)],style=POINT,symbolsize=20):
  display([A,B]);
```

$$p2 := -0.000004775210645\, x^{10} + 3.3\ 10^{-13}\, x^{9} + 1.000000001 + 0.0003330709441\, x^{8} + 1.5\ 10^{-9}\, x - 1.5\ 10^{-11}\, x^{7} - 0.4990604613\, x^{2} - 0.008540464293\, x^{6} - 9.9\ 10^{-10}\, x^{3} + 3.8\ 10^{-10}\, x^{5} + 0.09830882996\, x^{4}$$




```
> # esempio di instabilita nel calcolo N=30
  N   := 30 :
  a,b := -5, 5:
  X   := [seq(TR(a,b,zero2(N,i)),i=0..N)]:
  Y   := [seq(f(X[i+1]),i=0..N)]:
  X   := evalf(X,40):
  Y   := evalf(Y,40):
  p2  := interp(X,Y,x) ;

  A := plot([p2,f(x)],x=a..b,y=-1..2):
  B := plot([seq([X[i],Y[i]],i=1..N+1)],style=POINT,symbolsize=20):
  display([A,B]);
```

$$
\begin{aligned}
p2 := {} & 0.0001950869\, x - 0.0012407812\, x^{3} - 0.9737530154\, x^{2} - 0.000304094900\, x^{11} \\
& + 8.6100885\; 10^{-7}\, x^{17} - 0.0000086977564\, x^{15} + 0.000061902236\, x^{13} \\
& - 0.00001334510186\, x^{18} - 6.0494677\; 10^{-8}\, x^{19} + 0.01155623069\, x^{12} \\
& - 0.001663188082\, x^{14} + 0.0001740985804\, x^{16} - 0.05757549810\, x^{10} - 0.4862358800\, x^{6} \\
& + 0.2016098338\, x^{8} + 7.477132850\; 10^{-7}\, x^{20} + 2.99659080\; 10^{-9}\, x^{21} \\
& - 1.02251596\; 10^{-10}\, x^{23} - 3.024405886\; 10^{-8}\, x^{22} + 8.591505367\; 10^{-10}\, x^{24} \\
& - 1.625138583\; 10^{-11}\, x^{26} + 2.28583096\; 10^{-12}\, x^{25} + 1.837162578\; 10^{-13}\, x^{28} \\
& - 3.01190000\; 10^{-14}\, x^{27} + 0.9999999995 - 9.387066510\; 10^{-16}\, x^{30} \\
& + 1.77236059\; 10^{-16}\, x^{29} + 0.0023278484\, x^{5} + 0.8058129415\, x^{4} - 0.00202653891\, x^{7}
\end{aligned}
$$

$+ 0.00099311168\, x^9$




```
> # esempio di instabilita nel calcolo N=50
  N   := 50 :
  a,b := -5, 5:
  X   := [seq(TR(a,b,zero2(N,i)),i=0..N)]:
  Y   := [seq(f(X[i+1]),i=0..N)]:
  X   := evalf(X,40):
  Y   := evalf(Y,40):
  p2  := interp(X,Y,x) ;

  A := plot([p2,f(x)],x=a..b,y=-1..2):
  B := plot([seq([X[i],Y[i]],i=1..N+1)],style=POINT,symbolsize=20):
  display([A,B]);
```

$$p2 := 2.882249475\, x - 50.08697118\, x^3 - 0.9806176919\, x^2 - 861.5188370\, x^{11}$$
$$+ 92.76648485\, x^{17} - 261.5809509\, x^{15} + 556.0777981\, x^{13} + 4.43296653\, x^{18}$$
$$- 25.42063803\, x^{19} - 21.6344119\, x^{12} + 17.90202892\, x^{14} - 10.41293981\, x^{16}$$
$$+ 17.3546082\, x^{10} + 1.73809673\, x^6 - 8.3722067\, x^8 - 1.421491524\, x^{20}$$
$$+ 5.481709736\, x^{21} - 0.9426024220\, x^{23} + 0.3512762503\, x^{22} - 0.06796851310\, x^{24}$$
$$+ 6.503801890\; 10^{-21}\, x^{50} + 0.01043287226\, x^{26} + 0.1304942381\, x^{25}$$
$$- 0.001279635240\, x^{28} - 0.01463257499\, x^{27} + 0.0001262065583\, x^{30}$$
$$+ 0.001333824459\, x^{29} - 0.00009887933304\, x^{31} - 0.00001002816708\, x^{32}$$

$+ 261.1873948\, x^{5} + 0.000005946854947\, x^{33} + 0.605721080\, x^{4} + 6.410723227\ 10^{-7}\, x^{34}$
$- 2.885697530\ 10^{-7}\, x^{35} - 3.283336788\ 10^{-8}\, x^{36} + 1.118690071\ 10^{-8}\, x^{37}$
$- 3.411989206\ 10^{-10}\, x^{39} + 1.335297698\ 10^{-9}\, x^{38} + 7.998259284\ 10^{-12}\, x^{41}$
$- 4.249833190\ 10^{-11}\, x^{40} - 1.389321178\ 10^{-13}\, x^{43} + 1.034786752\ 10^{-12}\, x^{42}$
$+ 2.322268596\ 10^{-16}\, x^{46} - 1.268408199\ 10^{-17}\, x^{47} + 1.683041710\ 10^{-15}\, x^{45}$
$- 1.859528364\ 10^{-14}\, x^{44} - 1.798687335\ 10^{-18}\, x^{48} - 646.5638987\, x^{7}$
$+ 4.474747220\ 10^{-20}\, x^{49} + 1.000004331 + 927.4019173\, x^{9}$